Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 17

Double Integrals over Rectangular Regions

Initializations

```
> restart;
  with(plots):
  with(student):
```

17.1 Double Integrals over Rectangular Regions

Single integrals can be used to compute the area of a region under a curve and above a line segment on the x-axis. Similarly double integrals can compute the volume of a solid under a surface and above a region in the *xy*-plane.

Examples

Example 17.1.1

Compute the volume of the solid above the rectangle $\{(x, y) | 0 \le x \le 5, 0 \le y \le 7\}$ in the *xy*-plane and under the paraboloid $z = x^2 + 3y^2 + 4$.

Solution

It is always important to sketch the integration region. Make it a habit to provide your picture with a small margin.

```
> implicitplot({x=0, x=5, y=0, y=7}, x=-1..6, y=-1..8,
  axes=boxed);
```



We express the volume as a repeated integral: $\int_0^7 \int_0^5 (x^2 + 3y^2 + 4)\, dx\, dy$. To illustrate the procedure of repeated integration we first evaluate the "inner" integral (over *x*) and then the "outer" integral (over *y*).

```
> f:=(x, y)->x^2+3*y^2+4;
```

$$f := (x, y) \to x^2 + 3y^2 + 4 \qquad \textbf{(2.1.1.1)}$$

```
> int_x:=Int(f(x, y), x=0..5);
```

$$int\_x := \int_0^5 (x^2 + 3y^2 + 4)\, dx \qquad (2.1.1.2)$$

```
> int_x:=value(int_x);
```

$$int\_x := \frac{185}{3} + 15y^2 \qquad (2.1.1.3)$$

Observe that this is an expression in *y* which now needs to be integrated over the interval $[0, 7]$.

```
> int_y:=Int(int_x, y=0..7);
```

$$int\_y := \int_0^7 \left(\frac{185}{3} + 15y^2\right) dy \qquad (2.1.1.4)$$

```
> volume:=value(int_y);
```

$$volume := \frac{6440}{3} \qquad (2.1.1.5)$$

Usually such volume is expressed as a Double integral $\int_0^7 \int_0^5 (x^2 + 3y^2 + 4)\, dx\, dy$. Observe that the outer limits are associated with the variable *y* while the inner limits are associated with the variable *x*. We can evaluate an expression like the above directly by applying the **value** command.

```
> V:=Int(Int(x^2+3*y^2+4, x=0..5), y=0..7);
```

$$V := \int_0^7 \int_0^5 (x^2 + 3y^2 + 4)\, dx\, dy \qquad (2.1.1.6)$$

```
> test:=value(V);
```

$$test := \frac{6440}{3} \qquad (2.1.1.7)$$

The **student** package contains a **Doubleint** command which allows the user to code an iterated integral in a slightly more succinct form.

```
> alternative:=Doubleint(x^2+3*y^2+4, x=0..5, y=0..7);
```

$$alternative := \int_0^7 \int_0^5 (x^2 + 3y^2 + 4)\, dx\, dy \qquad (2.1.1.8)$$

```
> value(alternative);
```

$$\frac{6440}{3} \qquad (2.1.1.9)$$

Example 17.1.2

Integrate the function $f(x, y) = x\,y\sqrt{5x^2 + y^2}$ over the region bounded by: $x = -1$, $x = 5$, $y = 2$, and $y = 4$. Use two different approaches:
a) First integrate over *x*, then over *y*.
b) First integrate over *y*, then over *x*.

Solution
Code the function *f* and plot the integration region.

```
> f:=(x, y)->x*y*sqrt(5*x^2+y^2);
  implicitplot({x=-1, x=5, y=2, y=4}, x=-2..6, y=1..5,
  color=blue, axes=boxed);
```

$$f := (x, y) \to x\, y\, \sqrt{5\, x^2 + y^2}$$



a) First integrate over *x*, then over *y*.

```
> e1:=Int(Int(f(x, y), x=-1..5), y=2..4);
  e2:=value(e1);
  evalf(%);
```

$$e1 := \int_{2}^{4}\int_{-1}^{5} x\, y\, \sqrt{5\, x^2 + y^2}\, dx\, dy$$

$$e2 := \frac{81}{25} - \frac{5547}{25}\sqrt{3}\sqrt{43} - \frac{147}{25}\sqrt{3}\sqrt{7} + \frac{6627}{25}\sqrt{3}\sqrt{47}$$

$$603.872689 \qquad \textbf{(2.1.2.1)}$$

b) First integrate over *y*, then over *x*.

```
> e3:=Int(Int(f(x, y), y=2..4), x=-1..5);
  e4:=value(e3);
  evalf(%);
```

$$e3 := \int_{-1}^{5}\int_{2}^{4} x\, y\, \sqrt{5\, x^2 + y^2}\, dy\, dx$$

$$e4 := \frac{81}{25} - \frac{5547}{25}\sqrt{3}\sqrt{43} - \frac{147}{25}\sqrt{3}\sqrt{7} + \frac{6627}{25}\sqrt{3}\sqrt{47}$$

$$603.872689 \qquad \textbf{(2.1.2.2)}$$

```
>
```

Example 17.1.3

Use the Doubleint command in the student package to generate the integrals computed in the previous example.

Solution
Watch the reversal of order in the given code.

```
> First_integrate_over_x_then_over_y:=Doubleint(f(x, y), x=
```

```
-1..5, y=2..4);
```

$$First\_integrate\_over\_x\_then\_over\_y := \int_{2}^{4}\int_{-1}^{5} x\,y\,\sqrt{5\,x^{2}+y^{2}}\,\mathrm{d}x\,\mathrm{d}y \qquad \textbf{(2.1.3.1)}$$

```
> First_integrate_over_y_then_over_x:=Doubleint(f(x, y), y=
2..4, x=-1..5);
```

$$First\_integrate\_over\_y\_then\_over\_x := \int_{-1}^{5}\int_{2}^{4} x\,y\,\sqrt{5\,x^{2}+y^{2}}\,\mathrm{d}y\,\mathrm{d}x \qquad \textbf{(2.1.3.2)}$$

```
>
```